A Data User’s Guide to the BRFSS Sugar-Sweetened Beverage Questions:
How to Analyze Consumption of Sugar-Sweetened Beverages

Sohyun Park, PhD, and Liping Pan, MD, MPH

TABLE OF CONTENTS

SECTION I: Background
Introduction…………………………………………………………………………………………….2
Rationale…………………………………………………………………………………………….2

SECTION II: Using BRFSS to examine the frequency of sugar-sweetened beverage consumption
SSB Module……………………………………………………………………………………………3
Estimating the Prevalence of Sugar-Sweetened Beverage Consumption…………………...6

SECTION III: Selected Limitations to Using SSB Module Questions and Conclusions
Selected Limitations………………………………………………………………………………9
Conclusions………………………………………………………………………………………..9

APPENDIX
A. BRFSS 2011-2013 Sugar-Sweetened Beverage Optional Module Questions………10

B. Statistical Analysis Software (SAS) Callable-SUDAAN Code to Estimate the
Prevalence of Sugar-Sweetened Beverage Intake by Using BRFSS…………………..13

REFERENCES…………………………………………………………………………………….22
SECTION I. Background

1. Introduction

The BRFSS Sugar-Sweetened Beverage (SSB) Module, either included as an Optional Module or as State-Added Questions, can be used to monitor consumption of SSBs at the state level. The purpose of this guidance document is to provide a rationale for assessing SSB intake and to increase the capacity of data analysts to use the SSB Module. The BRFSS SSB Module has been offered as an Optional Module in 2011, 2012, and 2013.

This guidance document will provide
1. The rationale for collecting SSB intake data.
2. Guidance on how to use the 2011-2013 Sugar-Sweetened Beverage Optional Module to estimate prevalence of sugar-sweetened beverage consumption (daily, weekly, or monthly).

2. Rationale

The frequent consumption of SSB among adults has been associated with weight gain and obesity in longitudinal studies, but findings from cross-sectional studies are inconsistent.\(^1\)\(^-\)\(^3\) In addition to obesity, research also suggests that frequent SSB intake may be related to unhealthy dietary behavior\(^4\) and adverse health consequences, including type 2 diabetes,\(^1\)\(^3\) cardiovascular disease risk,\(^1\)\(^3\)\(^-\)\(^5\) and nonalcoholic liver disease.\(^6\) For some of these diseases, SSB consumption may work through obesity as a mediator, but for some the association appears to be independent of weight status.\(^1\)

The 2011-2012 National Health and Nutrition Examination Survey (NHANES) data showed that obesity levels are high as 34.9% of adults have obesity on the basis of measured weight and height data.\(^7\) On the basis of self-reported weight and height data, the prevalence of obesity varies by state, and it ranged from 21.3% to 35.1% in 2013.\(^8\) Excess caloric consumption contributes to obesity, and one of the significant contributors of calories to the US diet is SSBs,\(^9\) which is also a significant source of added sugars in the diet.\(^10\) The 2010 Dietary Guidelines for Americans define SSBs as, “Liquids that are sweetened with various forms of sugars that add calories. These beverages include, but are not limited to, soda, fruit ades and fruit drinks, and sports and energy drinks”.\(^11\) SSBs also tend to have few, if any, other nutrients.

On the basis of the 2009-2010 NHANES data, the mean energy intake from SSBs was 155 kcal (8.0% of total daily energy intake) for youth, and 151 kcal (6.9% of total daily energy intake) for
adults on any given day. The main consumers of SSBs were adolescents aged 12 to 19 years, male, non-Hispanic black, Mexican-American, and low-income individuals.

In addition to their contribution to caloric consumption, several other biological mechanisms have been proposed to explain the relationship between SSB consumption and obesity. First, when calories are consumed as liquids, people may not subsequently compensate their energy intake, resulting in extra consumption energy. Second, the prompt drop in blood sugar that follows the insulin response to intake of foods or beverages high in sugars, such as SSBs, increases hunger and may increase food intake. Third, fructose (a sugar found in commonly used sweeteners in the United States) does not activate hormones that help regulate satiety, which can result in not feeling full after consuming energy from SSBs.

Reasons for using the optional module include the following:

- The data can be used for surveillance of SSB intake frequency and for comparison of intake across states. Healthy People 2020 objectives include the reduction of calories from added sugars (objective number NWS-17.2) because SSB is the largest food category source of added sugars in the diet.
- The data can be used with other information to monitor the effect of statewide and state-supported interventions to prevent and control obesity, diabetes, and other chronic diseases.
- The data can be used to guide and take action about disparities in SSB intake. States can use their SSB intake data to identify characteristics of high SSB consumers and to develop targeted interventions for decreasing SSB intake, such as encouraging the consumption of healthier beverages (e.g., plain water) as part of an obesity and chronic disease prevention strategy.

Section II: Using BRFSS to examine the frequency of sugar-sweetened beverage consumption

1. SSB Module

The SSB Module was available as an Optional Module to all BRFSS states and territories during 2011-2013. The two SSB questions in 2011 and 2012 BRFSS, which were related to the consumption of regular soda and fruit drinks, were selected because they represented the most commonly consumed types of SSBs. On the basis of the 2009-2010 NHANES data, the mean energy intake of all SSBs among US adults was 151 kilocalories, with approximately 83 kcal/day from sodas and 29 kcal/day from fruit drinks.

One soda question and one fruit drink question originally were included as a BRFSS Optional Module in 2011 (see Appendix A). In 2012, the two questions were modified to add the “during
the past 30 days” time frame to the existing questions. The 2012 soda question was also included in the 2013 survey. However, the fruit drinks question was modified in 2013 to include sweet tea, sports drinks, and energy drinks, and this change was made to more accurately assess total SSB intake. In addition, different interviewer’s notes were added in the 2012 and 2013 surveys (see Appendix A). Six states used the SSB module in 2011 (Delaware, Hawaii, Iowa, Minnesota, New Jersey, and Wisconsin); 18 states used the module in 2012 (California, Delaware, Georgia, Hawaii, Iowa, Kansas, Maryland, Minnesota, Mississippi, Montana, Nebraska, Nevada, New Hampshire, New Jersey, New York, Oklahoma, South Dakota, and Tennessee); and 24 states used the module in 2013 (Alaska, Arizona, California, Connecticut, District of Columbia, Indiana, Iowa, Kansas, Kentucky, Louisiana, Maryland, Minnesota, Mississippi, Nebraska, New Jersey, New York, North Carolina, Ohio, Oklahoma, South Carolina, Utah, Vermont, West Virginia, and Wisconsin).

This document focuses on the analytic procedures to estimate the frequencies of SSB consumption. Additional BRFSS guidance documents provide technical assistance on the creation of one working analytical data set after combining data collected from landline and/or cellular versions that were used by the states, including landline and cellular telephone users, or landline only telephone users, which can be found at [http://www.cdc.gov/brfss/annual_data/2011/BRFSS2011_Analysis.pdf](http://www.cdc.gov/brfss/annual_data/2011/BRFSS2011_Analysis.pdf) for the 2011 BRFSS, [http://www.cdc.gov/brfss/annual_data/2012/pdf/AnalysisofModules_2012.pdf](http://www.cdc.gov/brfss/annual_data/2012/pdf/AnalysisofModules_2012.pdf) for the 2012 BRFSS, and [http://www.cdc.gov/brfss/annual_data/2013/pdf/AnalysisofModules_2013.pdf](http://www.cdc.gov/brfss/annual_data/2013/pdf/AnalysisofModules_2013.pdf) for the 2013 BRFSS.
Table 1. Comparison of the 2011-2013 BRFSS Sugar-Sweetened Beverage Optional Module Questions and Variable Names by Beverage Type

<table>
<thead>
<tr>
<th>Sugar-Sweetened Beverage Optional Module Questions and Variable Names</th>
<th>Survey Year</th>
<th>2011</th>
<th>2012</th>
<th>2013</th>
</tr>
</thead>
<tbody>
<tr>
<td><strong>Regular Soda</strong></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>About how often do you drink regular soda or pop that contains sugar? Do not include diet soda or diet pop.</td>
<td>2011</td>
<td>During the past 30 days, how often did you drink regular soda or pop that contains sugar? Do not include diet soda or diet pop.</td>
<td>Same question as 2012 [ssbsugar]&lt;sup&gt;a&lt;/sup&gt;</td>
<td></td>
</tr>
<tr>
<td>[ssbsugar]&lt;sup&gt;a&lt;/sup&gt;</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td><strong>Fruit drink</strong></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>About how often do you drink sweetened fruit drinks, such as Kool-aid™, cranberry juice cocktail, and lemonade? Include fruit drinks you made at home and added sugar to.</td>
<td>2011</td>
<td>During the past 30 days, how often did you drink sweetened fruit drinks, such as Kool-aid™, cranberry juice cocktail, and lemonade? Include fruit drinks you made at home and added sugar to.</td>
<td></td>
<td></td>
</tr>
<tr>
<td>[ssbfruit]&lt;sup&gt;a&lt;/sup&gt;</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td><strong>Fruit drink/sweet tea/sports drink/energy drink</strong></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>During the past 30 days, how often did you drink sugar-sweetened fruit drinks (such as Kool-aid™ and lemonade), sweet tea, and sports or energy drinks (such as Gatorade™ and Red Bull™)? Do not include 100% fruit juice, diet drinks, or artificially sweetened drinks.</td>
<td>2011</td>
<td></td>
<td></td>
<td>[ssbfrut2]&lt;sup&gt;a&lt;/sup&gt;</td>
</tr>
</tbody>
</table>

<sup>a</sup> BRFSS variable name.
2. Estimating the Prevalence of SSB Consumption

The frequency of regular soda intake and fruit drinks or fruit drinks/sweet tea/sports drinks/energy drink intake is needed to estimate consumption of SSBs by using the 2011-2013 BRFSS SSB Optional Module. The method to calculate SSB consumption follows the three steps described below, and the SAS Callable-SUDAAN programming code is provided in Appendix B.

Step 1: Calculate the frequency of regular soda intake.

BRFSS queries about how often regular soda or pop was consumed. This consumption can be reported by the respondent per day, per week, or per month.

Daily intake
The SAS code in Appendix B illustrates how to calculate daily regular soda intake. Weekly intake can be converted to daily intake by dividing the frequency of weekly soda consumption by 7. Monthly intake can be converted to daily soda intake by dividing the frequency of monthly consumption by 30.

Following is the sample SAS code for calculating daily soda intake by using 2011 and 2013 BRFSS (the variable name “ssbsugar” needs to be changed to “ssbsugr1” for 2012 data):

```
if 101<= ssbsugar <=199 then soda= ssbsugar -100;
if 201<= ssbsugar <=299 then soda=(ssbsugar -200)/7;
if 301<= ssbsugar <=399 then soda=(ssbsugar -300)/30;
if ssbsugar =888 then soda=0;
```

Weekly intake
Daily soda intake can be converted to weekly intake by multiplying the frequency of daily consumption by 7. Monthly soda intake can be converted to weekly intake by dividing the frequency of monthly consumption by 4.3.

The SAS code for calculating weekly intake is not included in Appendix B. However, you can use the following SAS code to replace the code for daily regular soda intake if you are interested in weekly intake in 2011 and 2013 (the variable name “ssbsugar” needs to be changed to “ssbsugr1” for 2012 data):

```
if 101<= ssbsugar <=199 then sodaweek= (ssbsugar -100)*7;
if 201<= ssbsugar <=299 then sodaweek= (ssbsugar -200)/7;
if 301<= ssbsugar <=399 then sodaweek= (ssbsugar -300)/4.3;
if ssbsugar =888 then sodaweek=0;
```
Monthly intake

Daily intake can be converted to monthly intake by multiplying the frequency of daily consumption by 30. Weekly intake can be converted to monthly intake by multiplying the frequency of weekly consumption by 4.3.

The SAS code for calculating monthly intake is not included in Appendix B. However, you can use the following SAS code to replace the code for daily regular soda intake if you are interested in monthly intake in 2011 and 2013 (the variable name “ssbsugar” needs to be changed to “ssbsugr1” for 2012 data):

```sas
if 101<= ssbsugar <=199 then sodamonth= (ssbsugar -100)*30;
if 201<= ssbsugar <=299 then sodamonth=( ssbsugar -200)*4.3;
if 301<= ssbsugar <=399 then sodamonth= ssbsugar -300;
if ssbsugar =888 then sodamonth=0;
```

Step 2: Calculate the frequency of fruit drink intake (2011 and 2012 BRFSS) or fruit drink/sweet tea/sports drink/energy drink intake (2013 BRFSS).

BRFSS queries about how often sweetened fruit drink or fruit drink/sweet tea/sports drink/energy drink was consumed. This consumption can be reported by the respondent per day, per week, or per month.

Daily intake

The SAS code in Appendix B illustrates how to calculate daily intake. Weekly intake can be converted to daily intake by dividing the frequency of weekly consumption by 7. Monthly intake can be converted to daily intake by dividing the frequency of monthly consumption by 30.

Following is the sample SAS code for calculating daily fruit drink intake using 2011 BRFSS (the variable name “ssbfruit” needs to be changed to “ssbfrut1” for 2012 data and to “ssbfrut2” for 2013 data):

```sas
if 101<= ssbfruit <=199 then fruitdr= ssbfruit -100;
if 201<= ssbfruit <=299 then fruitdr=(ssbfruit -200)/7;
if 301<= ssbfruit <=399 then fruitdr=(ssbfruit -300)/30;
if ssbfruit =888 then fruitdr=0;
```

Weekly intake

Daily intake can be converted to weekly intake by multiplying the frequency of daily consumption by 7. Monthly intake can be converted to weekly intake by dividing the frequency of monthly consumption by 4.3.
The SAS code for calculating weekly intake is not included in Appendix B. However, you can use the following SAS code to replace the code for daily intake if you are interested in weekly intake in 2011 (the variable name “ssbfruit” needs to be changed to “ssbfrut1” for 2012 data and to “ssbfrut2” for 2013 data):

```sas
if 101<= ssbfruit <=199 then fruitdrweek= (ssbfruit -100)*7;
if 201<= ssbfruit <=299 then fruitdrweek= ssbfruit -200;
if 301<= ssbfruit <=399 then fruitdrweek=(ssbfruit -300)/4.3;
if ssbfruit =888 then fruitdrweek=0;
```

**Monthly intake**

Daily intake can be converted to monthly intake by multiplying the frequency of daily consumption by 30. Weekly intake can be converted to monthly intake by multiplying the frequency of weekly consumption by 4.3.

The SAS code for calculating monthly intake is not included in Appendix B. However, you can use the following SAS code to replace the code for daily intake if you are interested in monthly intake in 2011 (the variable name “ssbfruit” needs to be changed to “ssbfrut1” for 2012 data and to “ssbfrut2” for 2013 data):

```sas
if 101<= ssbfruit <=199 then fruitdrmonth= (ssbfruit -100)*30;
if 201<= ssbfruit <=299 then fruitdrmonth= (ssbfruit -200)*4.3;
if 301<= ssbfruit <=399 then fruitdrmonth= ssbfruit -300;
if ssbfruit =888 then fruitdrmonth=0;
```

**Step 3:** Sum the frequencies obtained from Step 1 and Step 2 to obtain the total frequency of SSB intake.

Following are the sample SAS codes for calculating total SSB intake:

- **Daily SSB intake:** `ssbday= sum (soda, fruitdr)`
- **Weekly SSB intake:** `ssbweek=sum (sodaweek, fruitdrweek)`
- **Monthly SSB intake:** `ssbmonth=sum (sodamonth, fruitdrmonth)`

**Three things to remember when analyzing data from the SSB Module**

1) Skip Step 3 if you are only interested in either soda or fruit drink consumption (fruit drink/sweet tea/sport drink/energy drink intake in 2013), but not the combined measure.

2) Categorize frequency of SSB intake on the basis of project needs. The SAS code in Appendix B categorized SSB intake as ≥0 times/day to <1 time/day and ≥1 time/day. However, other categories may be more appropriate for your program.

3) Variable names for soda and fruit drink (fruit drink/sweet tea/sport drink/energy drink intake in 2013) can change by year. For example, the variable for soda intake was named “ssbsugar” in 2011 and 2013 and was named “ssbsugr1” in 2012; the variable
for fruit drink intake was named “ssbfruit” in 2011 and “ssbfrut1” in 2012; and the variable that represents fruit drink/sweet tea/sport drink/energy drink intake was named “ssbfrut2” in 2013. It is important to confirm the variable names in the annual code book (http://www.cdc.gov/brfss/annual_data/annual_data.htm) before analyzing data.

Section III. Limitations to Using SSB Module Questions and Conclusions

1. Limitations

There are limitations that an analyst should be aware of when using SSB data from the BRFSS. First, like all BRFSS information, the SSB information is self-reported, which could result in recall and social desirability response biases. Second, the two questions in 2012 and 2013 were slightly modified to add the “during the past 30 days” time frame to the existing questions. Third, in 2011 and 2012, only two types of SSB, regular soda and fruit drinks, were available in the Optional Module, limiting the ability to present frequency estimates of total SSB intake. In 2013, the fruit drink question was modified to include intake from sweet teas, sports drinks, and energy drinks, in addition to sweetened fruit drinks. Therefore, the estimates in 2011 and 2012 should be presented and interpreted differently than those in 2013. Fourth, it may not feasible to combine 2011-2013 data together for analyses because of the discrepancies on survey questions. Fifth, the frequency of SSB consumption was assessed and not the volume of SSB consumption. Lastly, because this is cross-sectional data, when examining the association between SSB consumption and behaviors or health conditions, causality cannot be assessed. It is possible that the diagnosis of disease or condition (e.g., diabetes, obesity) leads to changes in individual behaviors. For example, individuals with obesity or those who try to lose weight may reduce their SSB consumption to reduce their sugar or caloric intake. However, longitudinal research has consistently shown an association between frequent SSB consumption and weight gain or obesity, type 2 diabetes, and cardiovascular disease risk.1-3

2. Conclusions

This data analysis guidance document provides a step-by-step approach to estimate frequencies of SSB consumption (daily, weekly, or monthly) for the 2011-2013 SSB Optional Module and provides Statistical Analysis Software (SAS) Callable-SUDAAN codes. There are additional BRFSS guidance documents that provide technical assistance on the creation of one working analytical data set for combining questionnaire versions used by the states. The data analysis guides for the optional modules are available on the BRFSS website for 2011, 2012, and 2013 BRFSS data. The BRFSS SSB module provides valuable information that can be used for surveillance, program planning, and program evaluation.
Appendix A. BRFSS Sugar-Sweetened Beverage Optional Module Questions

2011 BRFSS OPTIONAL MODULE 4: SUGAR SWEETENED BEVERAGES

1. About how often do you drink regular soda or pop that contains sugar? Do not include diet soda or diet pop.
   (273-275)
   1 ___ Times per day
   2 ___ Times per week
   3 ___ Times per month
   8 8 8 Never
   7 7 7 Don’t know / Not sure
   9 9 9 Refused

2. About how often do you drink sweetened fruit drinks, such as Kool-aid™, cranberry, and lemonade? Include fruit drinks you made at home and added sugar to.
   (276-278)
   1 ___ Times per day
   2 ___ Times per week
   3 ___ Times per month
   8 8 8 Never
   7 7 7 Don’t know / Not sure
   9 9 9 Refused

2012 BRFSS OPTIONAL MODULE 5: SUGAR SWEETENED BEVERAGES

Now I would like to ask you some questions about sugary beverages.

Interviewer note: Please remind respondents to include regular soda that they mixed with alcohol.

1. During the past 30 days, how often did you drink regular soda or pop that contains sugar? Do not include diet soda or diet pop.
   (248-250)
   1 ___ Times per day
   2 ___ Times per week
   3 ___ Times per month
   Do not read:
   8 8 8 None
   7 7 7 Don’t know / Not sure
   9 9 9 Refused
2. During the past 30 days, how often did you drink sweetened fruit drinks, such as Kool-aid\textsuperscript{TM}, cranberry juice cocktail, and lemonade? Include fruit drinks you made at home and added sugar to.

(251-253)

Interviewer note: Fruit drinks are sweetened beverages that often contain some fruit juice or flavoring. Do not include 100% fruit juice, sweet tea, coffee drinks, sports drinks, or energy drinks.

1 _ _ Times per day
2 _ _ Times per week
3 _ _ Times per month

Do not read:
8 8 8 None
7 7 7 Don’t know / Not sure
9 9 9 Refused

2013 BRFSS OPTIONAL MODULE 5: SUGAR DRINKS

1. During the past 30 days, how often did you drink regular soda or pop that contains sugar? Do not include diet soda or diet pop.

(346-348)

Please read: You can answer times per day, week, or month: for example, twice a day, once a week, and so forth.

1 _ _ Times per day
2 _ _ Times per week
3 _ _ Times per month

Do not read:
8 8 8 None
7 7 7 Don’t know / Not sure
9 9 9 Refused

2. During the past 30 days, how often did you drink sugar-sweetened fruit drinks (such as Kool-aid\textsuperscript{TM} and lemonade), sweet tea, and sports or energy drinks (such as Gatorade\textsuperscript{TM} and Red Bull\textsuperscript{TM})? Do not include 100% fruit juice, diet drinks, or artificially sweetened drinks.
Please read: You can answer times per day, week, or month: for example, twice a day, once a week, and so forth.

(349-351)

1 _ _ Times per day
2 _ _ Times per week
3 _ _ Times per month

Do not read:

8 8 8 None
7 7 7 Don’t know / Not sure
9 9 9 Refused
Appendix B. Statistical Analysis Software (SAS) Callable-SUDAAN Code to Estimate the Prevalence of Sugar-sweetened Beverage Intake using BRFSS 2011-2013 Data

2011 Data

options nofmterr;
*IT IS OPTIONAL. SAS WILL NOT PRINT ERROR MESSAGE REGARDING FORMATTING;

libname brfss "\cdc\project\CCHP_NCCD_DNPAO_Data\dn national data\BRFSS\data";
*REPLACE WITH THE FILE AND LOCATION OF YOUR DATA;

*2011 DATA;

*SELECT STATES THAT USED 2011 SSB OPTIONAL MODULE IN THE COMBINED LANDLINE AND CELLULAR TELEPHONE SURVEY;
data llcp11;
  set brfss.llcp2011;
  where _state in (15,27,55);
  _finalWt = _LLCPWT;
run;

*SELECT STATES THAT USED 2011 SSB OPTIONAL MODULE IN THE LANDLINE TELEPHONE SURVEY;
data land11;
  set brfss.land2011;
  where _state in (10,19);
  _finalWt = _landwt;
run;

*SELECT STATES THAT USED 2011 SSB OPTIONAL MODULE IN THE LANDLINE TELEPHONE SURVEY VERSION 2;
data land11v2;
  set brfss.land11v2;
  where _state=34;
  _finalWt = _LndWtV2;
run;

*COMBINE THE DATASETS ABOVE AND SELECT THE VARIABLES TO BE USED FOR THIS ANALYSIS;
data combine11;
  set llcp11 land11 land11v2;
  year=1;
keep _state _ststr _psu age sex _prace_mrace race2 _racegr2
hispanc2 educa income2 pregnant height3 weight2 _bmi5
ssbsugar ssbfruit _finalwt year;
run;

*STEP 1. CALCULATING THE FREQUENCY OF DAILY SODA INTAKE;

data brfss11;
  set combine11;
  
  if 101<=ssbsugar<=199 then soda=ssbsugar-100;
  if 201<=ssbsugar<=299 then soda=(ssbsugar-200)/7;
  if 301<=ssbsugar<=399 then soda=(ssbsugar-300)/30;
  if ssbsugar=888 then soda=0;

  *CATEGORIZING REGULAR SODA INTAKE AS ≥1 TIME/DAY AND ≥0 TIMES/DAY TO <1 TIME/DAY;
  sodagp=3;
  if soda>=1 then sodagp=1;
  if 0<=soda<1 then sodagp=2;

*STEP 2. CALCULATING THE FREQUENCY OF DAILY FRUIT DRINK INTAKE;

  if 101<=ssbfruit<=199 then fruitdr=ssbfruit-100;
  if 201<=ssbfruit<=299 then fruitdr=(ssbfruit-200)/7;
  if 301<=ssbfruit<=399 then fruitdr=(ssbfruit-300)/30;
  if ssbfruit=888 then fruitdr=0;

  *CATEGORIZING FRUIT DRINK INTAKE AS ≥1 TIME/DAY AND ≥0 TIMES/DAY TO <1 TIME/DAY;
  fruitdrgp=3;
  if fruitdr>=1 then fruitdrgp=1;
  if 0<=fruitdr<1 then fruitdrgp=2;

*STEP 3. SUMMARIZE THE FREQUENCIES OBTAINED FROM STEP 1 AND STEP 2 TO GET THE TOTAL FREQUENCY OF SSB INTAKE;
  if soda ne . and fruitdr ne . then
  ssb=sum(soda,fruitdr);
*CATEGORIZING TOTAL SSB INTAKE AS ≥1 TIME/DAY AND ≥0 TIMES/DAY TO <1 TIME/DAY;
  ssbgp=3;
  if ssb>=1 then ssbgp=1;
  if 0<=ssb<1 then ssbgp=2;
run;

proc sort;
  by _ststr _psu;
run;

*ESTIMATING PREVALENCE OF SSB INTAKE AT LEAST 1 TIME PER DAY BY STATE IN 2011;

proc descript filetype=sas design=wr data=work.brfss11;
nest _ststr _psu/ missunit;
subpopn sodagp!=3 and fruitdrgp!=3;
weight _finalWt;
var ssbgp;
catlevel 1;
tables _state;
subgroup _state;
levels 78;
print nsum wsum percent sepercent lowpct uppct total/percentfmt=F5.1
  sepercentfmt=F5.1 wsumfmt=F12.0 totalfmt=F12.0 lowpctfmt=F5.1 uppctfmt=F5.1
style=NCHS;
title "Prevalence of SSB intake at least 1 time per day, by state, 2011 BRFSS";
run;

2012 Data

*2012 DATA;

*SELECT STATES THAT USED 2012 SSB OPTIONAL MODULE IN THE COMBINED LANDLINE AND CELLULAR TELEPHONE SURVEY;
data llcp12;
  set brfss.llcp2012;
  where _state in (10, 13, 15, 27, 28, 30, 32, 33, 36, 46, 47);
  _finalWt = _llcpwt;
run;

*SELECT STATES THAT USED 2012 SSB OPTIONAL MODULE IN THE COMBINED LANDLINE AND CELLULAR TELEPHONE SURVEY VERSION 1;
data llcp12v1;
   set brfss.llcp12v1;
   where _state in (6, 19, 20, 40);
   _finalwt = _lcpwtv1;
run;

*SELECT STATES THAT USED 2012 SSB OPTIONAL MODULE IN THE COMBINED LANDLINE AND CELLULAR TELEPHONE SURVEY VERSION 2;
data llcp12v2;
   set brfss.llcp12v2;
   where _state in (24, 31, 34);
   _finalWt = _lcpwtv2;
run;

*COMBINE THE DATASETS ABOVE AND SELECT THE VARIABLES TO BE USED FOR THIS ANALYSIS;
data combine12;
   set llcp12 llcp12v1 llcp12v2;
   year=2;
   keep _state _ststr _psu age sex _prace _mrace race2 _racegr2
       hispanc2 educa income2 pregnant height3 weight2 _bmi5
       ssbsugr1 ssbfrut1 _finalwt year;
run;

*STEP 1. CALCULATING THE FREQUENCY OF DAILY SODA INTAKE;

data brfss12;
   set combine12(rename=(ssbsugr1= ssbsugar) rename=(ssbfrut1= ssbfruit));
   if 101<=ssbsugar<=199 then soda=ssbsugar-100;
   if 201<=ssbsugar<=299 then soda=(ssbsugar-200)/7;
   if 301<=ssbsugar<=399 then soda=(ssbsugar-300)/30;
   if ssbsugar=888 then soda=0;

*CATEGORIZING REGULAR SODA INTAKE AS ≥1 TIME/DAY AND ≥0 TIMES/DAY TO <1 TIME/DAY;
sodagp=3;
if soda>=1 then sodagp=1;
if 0<=soda<1 then sodagp=2;

*STEP 2. CALCULATING THE FREQUENCY OF DAILY FRUIT DRINK INTAKE;

    if 101<=ssbfruit<=199 then fruitdr=ssbfruit-100;
    if 201<=ssbfruit<=299 then fruitdr=(ssbfruit-200)/7;
    if 301<=ssbfruit<=399 then fruitdr=(ssbfruit-300)/30;
    if ssbfruit=888 then fruitdr=0;

*CATEGORIZING FRUIT DRINK INTAKE AS ≥1 TIME/DAY AND ≥0 TIMES/DAY TO <1 TIME/DAY;
    fruitdrgp=3;
    if fruitdr>=1 then fruitdrgp=1;
    if 0<=fruitdr<1 then fruitdrgp=2;

*STEP 3. SUMMARIZE THE FREQUENCIES OBTAINED FROM STEP 1 AND STEP 2 TO GET THE TOTAL FREQUENCY OF SSB INTAKE;
    if soda ne . and fruitdr ne . then
        ssb=sum(soda,fruitdr);

*CATEGORIZING TOTAL SSB INTAKE AS ≥1 TIME/DAY AND ≥0 TIMES/DAY TO <1 TIME/DAY;
    ssbgp=3;
    if ssb>=1 then ssbgp=1;
    if 0<=ssb<1 then ssbgp=2;

run;

proc sort;
    by _ststr _psu;
run;

*ESTIMATING PREVALENCE OF SSB INTAKE AT LEAST 1 TIME PER DAY BY STATE IN 2012;

proc descript filetype=sas design=wr data=work.brfss12;
nest _ststr _psu/ missunit;
subpopn sodagp!=3 and fruitdrgp!=3;
weight _finalWt;
var ssbgp;
catlevel 1;
tables _state;
subgroup _state;
levels 78;
print nsum wsum percent sepercent lowpct uppct total/percentfmt=F5.1
       sepercentfmt=F5.1 wsumfmt=F12.0 totalfmt=F12.0 lowpctfmt=F5.1 uppctfmt=F5.1
style=NCHS;
title "Prevalence of SSB intake at least 1 time per day, by state, 2012 BRFSS";
run;

2013 Data
*2013 DATA;

*SELECT STATES THAT USED 2013 SSB OPTIONAL MODULE IN THE COMBINED LANDLINE AND CELLULAR TELEPHONE SURVEY;
data llcp13;
   set brfss.llcp2013;
   where _state in (2, 4, 9, 11, 21, 22, 24, 27, 28, 45, 49, 50, 54, 55);
   _finalWt = _llcpwt;
run;

*SELECT STATES THAT USED 2013 SSB OPTIONAL MODULE IN THE COMBINED LANDLINE AND CELLULAR TELEPHONE SURVEY VERSION 1;
data llcp13v1a;
   set brfss.llcp13v1;
   where _state in (18, 19, 20, 31, 39);
   _finalwt = _lcpwtv1;
run;

data llcp13v1b;
   set brfss.llcp13v1;
   where _state in (6);
   _finalwt = _lcpwtv1/2;
run;

*SELECT STATES THAT USED 2013 SSB OPTIONAL MODULE IN THE COMBINED LANDLINE AND CELLULAR TELEPHONE SURVEY VERSION 2;
data llcp13v2a;
set brfss.llcp13v2;
   where _state in (36, 37, 40);
   _finalWt = _lcpwtv2;
run;

data llcp13v2b;
   set brfss.llcp13v2;
   where _state in (6);
   _finalWt = _lcpwtv2/2;
run;

*SELECT STATES THAT USED 2013 SSB OPTIONAL MODULE IN THE COMBINED LANDLINE AND CELLULAR TELEPHONE SURVEY VERSION 3;
data llcp13v3;
   set brfss.llcp13v3;
   where _state in (34);
   _finalWt = _lcpwtv3;
run;

*COMBINE THE DATASETS ABOVE AND SELECT THE VARIABLES TO BE USED FOR THIS ANALYSIS;
data combine13;
   set llcp13 llcp13v1a llcp13v1b llcp13v2a llcp13v2b llcp13v3;
   year=2;
   keep _state _ststr _psu _age80 sex _prace1 _mrace1 _race_g1 _race _hispanc educa income2 pregnant height3 weight2 _bmi5 ssbsugar ssbfrut2 _finalwt year;
run;

*STEP 1. CALCULATING THE FREQUENCY OF DAILY SODA INTAKE;
data brfss13;
   set combine13;
   if 101<=ssbsugar<=199 then soda=ssbsugar-100;
   if 201<=ssbsugar<=299 then soda=(ssbsugar-200)/7;
   if 301<=ssbsugar<=399 then soda=(ssbsugar-300)/30;
   if ssbsugar=888 then soda=0;
*CATEGORIZING REGULAR SODA INTAKE AS ≥1 TIME/DAY AND ≥0 TIMES/DAY TO <1 TIME/DAY;
    sodagp=3;
    if soda>1 then sodagp=1;
    if 0<=soda<1 then sodagp=2;

*STEP 2. CALCULATING THE FREQUENCY OF DAILY CONSUMPTION OF FRUIT DRINKS, SWEET TEA,
    SPORTS DRINKS, AND ENERGY DRINKS;
    if 101<=ssbfrut2<=199 then combinedr=ssbfrut2-100;
    if 201<=ssbfrut2<=299 then combinedr=(ssbfrut2-200)/7;
    if 301<=ssbfrut2<=399 then combinedr=(ssbfrut2-300)/30;
    if ssbfrut2=888 then combinedr=0;

*CATEGORIZING CONSUMPTION OF FRUIT DRINKS, SWEET TEA, SPORTS DRINKS,
    AND ENERGY DRINKS AS ≥1 TIME/DAY AND ≥0 TIMES/DAY TO <1 TIME/DAY;
    combdrgp=3;
    if combinedr>=1 then combdrgp=1;
    if 0<=combinedr<1 then combdrgp=2;

*STEP 3. SUMMARIZE THE FREQUENCIES OBTAINED FROM STEP 1 AND STEP 2 TO GET THE TOTAL FREQUENCY OF SSB INTAKE;
    if soda ne . and fruitdr ne . then
        ssb=sum(soda,combinedr);

*CATEGORIZING TOTAL SSB INTAKE AS ≥1 TIME/DAY AND ≥0 TIMES/DAY TO <1 TIME/DAY;
    ssbgp=3;
    if ssb>=1 then ssbgp=1;
    if 0<=ssb<1 then ssbgp=2;

run;

proc sort;
    by _ststr _psu;
run;

*ESTIMATING PREVALENCE OF SSB INTAKE AT LEAST 1 TIME PER DAY BY STATE IN 2013;
proc descript filetype=sas design=wr data=work.brfss13;
  nest _ststr _psu/ missunit;
  subpopn sodagp!=3 and combdrgp!=3;
  weight _finalWt;
  var ssbgp;
  catlevel 1;
  tables _state;
  subgroup _state;
  levels 78;
  print nsum wsum percent sepercent lowpct uppct total/percentfmt=F5.1
            sepercentfmt=F5.1 wsumfmt=F12.0 totalfmt=F12.0 lowpctfmt=F5.1
            uppctfmt=F5.1 style=NCHS;
  title "Prevalence of SSB intake at least 1 time per day, by state, 2013 BRFSS";
run;
REFERENCES


4. Sharkey JR, Johnson CM, Dean WR. Less-healthy eating behaviors have a greater association with a high level of sugar-sweetened beverage consumption among rural adults than among urban adults. *Food Nutr Res.* 2011;55.


